

## **Noções Básicas de Cálculo Diferencial e Integral com o Maple**

Angela Mallmann Wendt, Fabrício Fernando Halberstadt, Fernanda Ronssani de Figueiredo<sup>1</sup> e Lauren Maria Mezzomo Bonaldo

PET Matemática, Curso de Matemática, Centro de Ciências Naturais e Exatas, UFSM, RS

<sup>1</sup> fernanda.ronssani@gmail.com

Antonio Carlos Lyrio Bidel

Curso de Matemática, Centro de Ciências Naturais e Exatas, UFSM, RS

bidelac@gmail.com

### **RESUMO**

O estudo gráfico de elementos e situações-problema do Cálculo Diferencial e Integral nos cursos de graduação de matemática, por vezes, resume-se a poucos exemplos, devido às dificuldades que se encontram em desenhar manualmente funções mais complexas. Para isso, pretende-se nesse minicurso apresentar de forma sucinta os principais comandos e funções disponibilizados pelo software Maple, amplamente utilizado na visualização gráfica de funções matemáticas. Além disso, apresentar alguns problemas e discutir a sua interpretação gráfico-geométrica. Iniciaremos dando foco à plotagem e animação de gráficos de funções diversas no Maple em duas e em três dimensões. Nesta primeira parte, pretende-se trabalhar com funções clássicas como as funções seno, cosseno, cardióide, e curvas de nível, entre outras. Em seguida, serão estudados os comandos básicos para o cálculo e a plotagem de derivadas de funções bem como a interpretação geométrica desses cálculos. Num terceiro e último momento, serão abordados cálculos de integrais (definidas e indefinidas) simples, duplas e triplas no software.

**Palavras-chave:** Maple, Cálculo Diferencial e Integral, Gráficos, Derivadas, Integrais.

## *INTRODUÇÃO*

Este minicurso foi desenvolvido pelos bolsistas do Grupo PET (Programa de Educação Tutorial) Matemática Angela Mallmann Wendt, Fabrício Fernando Halberstadt, Fernanda Ronssani de Figueiredo e Lauren Maria Mezzomo Bonaldo, sob orientação do Professor Tutor Antonio Carlos Lyrio Bidel, como uma proposta de qualificar a formação de bolsistas e acadêmicos na utilização de novas tecnologias aplicadas ao ensino e aprendizagem da matemática.

Neste minicurso serão abordados comandos básicos do MAPLE que podem ser utilizados no cálculo de limites, derivadas e integrais, plotagem de gráficos bidimensionais e tridimensionais, bem como animações para os mesmos.

Esta apostila contempla de forma sucinta e introdutória os principais recursos do MAPLE, uma vez que serve de apoio didático de um minicurso de curta duração.

## 1. COMANDOS BÁSICOS

No MAPLE, para realizar as operações básicas adição, subtração, multiplicação e divisão são usados os comandos +, -, \* e /, respectivamente. De maneira geral, o Maple trabalha com números exatos na forma racional.

>(37\*5+13/7)^2;

$$\frac{1710864}{49}$$

Para obtermos este resultado na sua forma decimal utiliza-se o comando *evalf*.

>(37\*5+13/7)^2;

$$\frac{1710864}{49}$$

>evalf(%);

34915.591836734693878

>evalf[40](%%);

34915.59183673469387755102040816326530612

Observe que em *evalf [40](%%)* o número entre colchetes é o número de casas decimais. A quantidade de sinais de porcentagem remete ao cálculo efetuado anteriormente, por exemplo *evalf [40](%%)* representa a forma decimal do cálculo realizado há dois cálculos anteriores em  $(37*5+13/7)^2$ .

Outra forma de trabalhar com números decimais é a colocação de um ponto após um dos números constantes na operação.

>1./3;

0.3333333333

>(37\*5+13./7)^2;

34915.59185

Para um número maior de casas decimais, deve-se defini-lo através do comando *Digits*.

>Digits := 20;

*Digits:=20*

>(37\*5+13./7)^2;

34915.591836734693876

### 1.1. Algumas funções matemáticas no maple

Abaixo apresentamos uma pequena lista da sintaxe de algumas funções matemáticas no Maple:

- funções trigonométricas:  $\sin(x)$ ,  $\cos(x)$ ,  $\tan(x)$ ,  $\cot(x)$ ,  $\sec(x)$ ,  $\csc(x)$ ;
- funções trigonométricas inversas:  $\arcsin(x)$ ,  $\arccos(x)$ ,  $\arctan(x)$ ,  $\text{arccot}(x)$ ,  $\text{arcsec}(x)$ ,  $\text{arccsc}(x)$ ;
- função exponencial de base e:  $\exp(x)$ ;
- função logarítmica de base e:  $\ln(x)$ ;
- função logarítmica de base a, sendo  $a > 0$  qualquer:  $\log[a](x)$ ;

- funções hiperbólicas:  $\sinh(x)$ ,  $\cosh(x)$ ,  $\tanh(x)$ ,  $\operatorname{sech}(x)$ ,  $\operatorname{csch}(x)$ ,  $\operatorname{coth}(x)$ ;
- funções hiperbólicas inversas:  $\operatorname{arcsinh}(x)$ ,  $\operatorname{arccosh}(x)$ ,  $\operatorname{arctanh}(x)$ ,  $\operatorname{arcsech}(x)$ ,  $\operatorname{arccsch}(x)$ ,  $\operatorname{arccotgh}(x)$ .

Para obter a lista completa de funções trigonométricas digite `>?inifcn`.

### 1.2. Funções

Podemos definir uma função no Maple por meio do sinal `->` (sinal de menos seguido do sinal de maior).

`>f:=x->2*x^2-7;`

$$f:=x \rightarrow 2x^2 - 7$$

`>f(13);`

$$331$$

### 1.3. Atribuindo nomes

Como muitas vezes é preciso realizar cálculos extensos, é necessário nomear as equações. Para isso, utiliza-se “:=”.

`>equacao:=2*x^2+5*x-3=0;`

$$equacao=2x^2+5x-3=0$$

Para encontrar as raízes da equação usa-se o comando `solve`.

`>solve(equacao);`

$$\frac{1}{2}, -3$$

Podemos também nomear as soluções por meio dos comandos `nome da primeira solução:= %[1];` e `nome da segunda solução:= %%[2];`.

`>x1:= %[1];`

$$x1:=\frac{1}{2}$$

`>x2:= %%[2];`

$$x2:=-3$$

Note que somente o sinal de igualdade não define uma função, ela não muda o valor da variável.

`>y=x+2;`

$$y=x+2$$

`>y;`

$$y$$

Mas com o comando “:=” temos:

`>y:=x+2;`

$$y:=x+2$$

`>y;`

$$x+2$$

Quando não quisermos mais trabalhar com a atribuição que fizemos, é só utilizar


```
>y:='y';
```

```
y:=y
```


## 2. GRÁFICOS

O comando para traçar gráficos é o *plot* sendo que a sua forma geral é *plot(f(x), x=a..b, y=c..d, opções)*; na qual x indica o intervalo das abscissas e y o intervalo das ordenadas. Em opções define-se o estilo da visualização gráfica. Pode-se definir uma cor pelo comando *color* (podendo ser: *aquamarine, black, blue, navy, coral, cyan, brown, gold, green, gray, grey, khaki, magenta, maroon, orange, pink, plum, red, sienna, tan, turquoise, violet, wheat, white, yellow*), *thickness* (espessura da linha) e *style* (tipo de linha, podendo ser: *point, patch, line, patchnogrid, polygonoutline, polygon* e *default*).

```
>plot(x^3, x=-3..3, y=-2..2, color=green, thickness=5, style=line);
```




```
>plot(sin(x), x=-2..2, y=-2..2, color=blue, style=point, numpoints=100);
```




### 2.1. Escala

Utiliza-se *constrained* (ambos os eixos com a mesma escala) ou *unconstrained* (os eixos não possuem necessariamente a mesma escala).

```
> plot(x^2+1,x=-4..4,y=0..16,scaling=constrained);
```




```
>plot(x^2+1,x=-4..4,y=0..16,scaling=unconstrained);
```



## 2.2. Outros recursos


O Maple disponibiliza também recursos auxiliares na visualização dos gráficos.

```
>plot(arctan(t),t=-2..2,color=green,thickness=5,title= `Gráfico da função arctangente`);
```



Podemos inserir mais de uma função em um mesmo gráfico.

```
>plot([exp(x^2),1+x^3,1+x+x^2/3],x=-1..1,color=[black,red,green],thickness=3,linestyle=[DASHDOT,DASH,SOLID]);
```



Observação: Para trocar de linha sem executar o comando, é necessário que se tecle SHIFT + ENTER.

> restart:with(plots):


g:=plot([x^2,x^3,x+2],x=-4..4,y=-10..10,thickness=3,color=[blue,red,green],title='Funções');

t1:=textplot([-2.4,3.2,'Função Quadrática']):

t2:=textplot([-1.75,-5.2,'Função Cúbica']):

t3:=textplot([4,5,'Função linear']):


display([g,t1,t2,t3]);



### 2.3. Funções Parametrizadas

O comando para plotar funções parametrizadas é:  $\text{plot}([x(t), y(t), t=a..b], \text{opções})$ ; abaixo construímos o gráfico da curva paramétrica definida por:  $x(t) = t \cos(2\pi/t)$ ,  $y(t) = t \sin(2\pi/t)$ ,  $t = [0, 5]$ .

> plot([t\*cos(2\*Pi/t), t\*sin(2\*Pi/t), t=0..5], thickness=5, color=blue, scaling=constrained);




### 2.4. Coordenadas Polares

A forma para plotar gráficos de funções em coordenadas polares é  $\text{polarplot}(r(\theta), \theta=a..b, \text{opções})$  ou  $\text{plot}([x(t), y(t), t=a..b], \text{coords=polar})$ .

Abaixo temos o gráfico de uma rosácea de oito pétalas:

> plot(-16\*cos(4\*theta), theta=0..2\*Pi, coords=polar, thickness=5, color=pink, title= Rosácea de oito Pétalas);



## 2.5. Animação de gráficos de uma variável


O comando para animação de gráficos é: `plots[animate](f(x,t), x=a..b, t=c..d, frames=n)`; onde  $f(x,t)$  é uma expressão (ou função) de duas variáveis.

A variação de  $x$  corresponde ao domínio das funções envolvidas na animação, enquanto que a variação do  $t$  corresponde às posições intermediárias.


O valor  $t=c$  corresponde ao gráfico inicial e  $t=d$  corresponde ao gráfico final. O total de  $n$  gráficos construídos é controlado com a opção `frames = n`.

>with(plots):

>animate(x^2+t, x=-3..3, t=0..12, frames=12);



>plots[animate](theta/t, theta=0..8\*Pi, t=1..4, coords=polar, numpoints=200);



## 2.6. Gráficos de duas Variáveis

Para plotar um gráfico de uma função  $f(x,y)$  de duas variáveis é necessária a utilização do comando `plot3d`, o qual possui algumas variações em sua sintaxe dependendo do que se deseja traçar. A sintaxe básica é a seguinte: `plot3d(f(x,y), x=a..b, y=c..d, opções)`; onde os parâmetros " $f(x,y)$ ", " $x=a..b$ " e " $y=c..d$ " são obrigatórios enquanto que o parâmetro "`opções`" é opcional.




### 2.6.1. Alguns Comandos


Os gráficos podem ser personalizados também com os comandos: *grid*=[*m,n*] usado para refinar o desenho do gráfico, *m* é o número de pontos na direção da primeira coordenada, e *n* no da segunda coordenada; *style*, que pode ser: *point*, *hidden*, *patch*, *wireframe*, *contour*, *patchnograd*, *patchcontour*, *surface*, *surfacecontour*, *surfacewireframe*, *wireframeopaque* ou *line*.

Exemplos:

```
> plot3d(sin(x*y), x=-5..5, y=-5..5, grid=[30,30], style=line, color=blue);
```




```
> c1:= [cos(x)-2*sin(0.4*y),sin(x)-2*sin(0.4*y),y]:
c2:= [cos(x)+2*sin(0.4*y),sin(x)+2*sin(0.4*y),y]:
c3:= [cos(x)+2*cos(0.4*y),sin(x)-2*cos(0.4*y),y]:
c4:= [cos(x)-2*cos(0.4*y),sin(x)+2*cos(0.4*y),y]:
plot3d({c1,c2,c3,c4},x=0..2*Pi,y=0..10,grid=[25,15],style=line,color=green);
```



```
> sphereplot(1 + theta + phi, theta=0..2*Pi, phi=0..Pi, style=line, thickness=3);
```



```
> cylinderplot(3*theta + 2, theta=0..2*Pi, z=-5..5, color=magenta, thickness=3, style=line);
```




```
>implicitplot3d((x^2+(9/4)*y^2+z^2-1)^3-x^2*z^3-(9/80)*y^2*z^2=0, x=-1.5..1.5, y=-1.5..1.5, z=-1.5..1.5,color=red,style=patch,numpoints=90000);
```




## 2.7. Animação de gráficos de duas variáveis

Exemplos:

```
> animate3d(x*sin(t*u),x=1..3,t=1..4,u=2..4);
```




```
> animate3d(cos(t*x)*sin(t*y),x=-Pi..Pi, y=-Pi..Pi,t=1..2,style=line,thickness=2,color=cyan,coords=cylindrical);
```




## 2.8. Curvas de nível

Exemplos:


```
> contourplot(sin(x^3-y),x=-10..10,y=-10..10,contours=3,filled=true);
```




```
> plot3d(sin(x^3-y),x=-10..10,y=-10..10, style=contour, contours=3,filled=true);
```



```
> contourplot(x^3+y^3-x^2+y^2+x-5,x=-10..10,y=-10..10,color=blue,contours=300);
```



```
>plot3d(x^3+y^3-x^2+y^2+x-5,x=-10..10,y=-10..10,color=blue, style=contour, contours=300);
```



### 3. LIMITES

Para calcular o limite de uma função quando a variável tende a certo valor, é necessário utilizar o comando *limit*. Por exemplo: *limit(f(t), t=a)*, onde *a* é a variação. *Limit* é utilizado para deixar indicado o limite, já o comando *limit* é utilizado para resolver o limite. O uso do *Limit* combinado com o *limit* pode melhorar a apresentação do resultado.

```
>Limit(cos(a*x)/(b*x), x=1);
```

$$\lim_{x \rightarrow 1} \frac{\cos(ax)}{bx}$$

```
>limit(cos(a*x)/(b*x), x=1);
```

$$\frac{\cos(a)}{b}$$

```
>Limit(cos(a*x)/(b*x), x=1)=limit(cos(a*x)/(b*x), x=1);
```

$$\lim_{x \rightarrow 1} \frac{\cos(ax)}{bx} = \frac{\cos(a)}{b}$$

#### 3.1. Limite de funções

```
>Limit((x^3+5*x^2)/(x^4+x^5),x=0)=limit((x^3+5*x^2)/(x^4+x^5),x=0);
```

$$\lim_{x \rightarrow 0} \frac{x^3 + 5x^2}{x^4 + x^5} = \infty$$

#### 3.2. Limites Laterais

Para calcular limites laterais acrescenta-se uma opção *left* ou *right* aos comandos *limit* e ou *Limit*. Se for acrescentada a opção *left*, então, será calculado o limite lateral à esquerda. Se for acrescentado *right*, então o limite será lateral à direita.

```
>Limit(cos(Pi*x)/x, x=0, left)=limit(cos(Pi*x)/x, x=0, left);
```

$$\lim_{x \rightarrow 0^-} \frac{\cos(\pi x)}{x} = -\infty$$

>Limit(cos(Pi\*x)/x, x=0, right)= limit(cos(Pi\*x)/x, x=0, right);

$$\lim_{x \rightarrow 0^+} \frac{\cos(\pi x)}{x} = \infty$$

>Limit(cos(Pi\*x)/x, x=0)= limit(cos(Pi\*x)/x, x=0);

$$\lim_{x \rightarrow 0} \frac{\cos(\pi x)}{x} = \text{undefined}$$

### 3.3. Pontos de Descontinuidade

Para calcular o limite de funções não contínuas devemos utilizar os limites laterais.

>f:=piecewise(x<2,3\*x,x>=2,x^3);

$$f := \begin{cases} 3x & x < 2 \\ x^3 & 2 \leq x \end{cases}$$

>limit(f,x=2,left);

6


>limit(f,x=2,right);

8

>limit(f,x=2);

undefined

>plot(f,x=-0..3);



### 3.4. Limites no infinito

Para calcular limites no infinito, isto é, com a variável tendendo a  $+\infty$  ou  $-\infty$ , utilizamos *-infinity* ou *infinity* para a variável.

>Limit(1/x)\*sin(x), x=infinity)=limit(1/x)\*sin(x), x=infinity);

$$\lim_{x \rightarrow \infty} \frac{\sin(x)}{x} = 0$$

>Limit(x^3+x^2-3, x=-infinity)=limit(x^3+x^2-3, x=-infinity);

$$\lim_{x \rightarrow (-\infty)} x^3 + x^2 - 3 = -\infty$$

## 4. DERIVADAS

Para calcular derivadas utiliza-se basicamente o comando  $\text{diff}(f,x)$ .

Exemplo:

>f:=x^3+tan(x)+100;

$$f := x^3 + \tan(x) + 100$$

>diff(f,x);

$$3x^2 + 1 + \tan(x)^2$$

### 4.1. Derivação de ordem $n$ em relação a uma variável

Para calcular este tipo de derivada o comando é praticamente o mesmo, pois a ordem da derivada está relacionada com o número de “x” que atribuímos dentro do comando  $\text{diff}(f,x)$ . Por exemplo: se quisermos saber qual é a derivada terceira de uma função  $f(x)$ , basta colocarmos “três  $x_s$ ” no comando, ou seja,  $\text{diff}(f,x,x,x)$ . Portanto o comando para derivação de ordem  $n$  em relação a uma variável é  $\text{diff}(f,x,x,\dots,x)$ . Para simplificar o comando  $\text{diff}(f,x,x,\dots,x)$ , basta colocar  $\text{diff}(f,x\$n)$ .

Exemplos:

>f:= x^5+ x^3+ exp(x);

$$f := x^5 + x^3 + e^x$$

>diff(f,x);

$$5x^4 + 3x^2 + e^x$$

>diff(f,x,x);

$$20x^3 + 6x + e^x$$

>diff(f,x,x,x);

$$60x^2 + 6 + e^x$$

>diff(f,x\$4);

$$120x + e^x$$

>diff(f,x\$9);

$$e^x$$

>diff(f,x\$15);


$$e^x$$

### 4.2. Reta tangente

Alguns exemplos com retas tangentes:

>with(student):

```
>showtangent(x^2+12,x=6);
```



Analisemos como se comporta a reta tangente a um ponto em relação à função nesse ponto.

```
>with(plots):
```

```
>g:=x->-x^2+5*x;
```

$$g := x \rightarrow -x^2 + 5x$$


```
>a := plot(g(x), x=0..5, color = green):
```

```
>p := [[1, g(1)]];
```

$$p := [[1, 4]]$$

```
>b := plot(p, x=0..5, style=point, symbol=diamond, color=black):
```


```
>display([a, b]);
```




```
>with(student):
```

```
>c := showtangent(g(x),x=1, x=0..6):
```


```
> display(a, b, c);
```




>showtangent(g(x), x=1, x=0.5..2);



>showtangent(g(x), x=1, x=0.8..1.2);



>showtangent(g(x), x=1, x=0.99..1.1);





## 5. INTEGRAIS

O Maple possui diversos recursos para o cálculo de integrais indefinidas, definidas e impróprias.

### 5.1. Integrais de funções

A integral (primitiva) de uma função definida por uma expressão algébrica  $f(x)$  é calculada com o comando  $\text{int}(f(x), x)$ . Esse comando também possui uma forma inercial:  $\text{Int}(f(x), x)$ . A forma inercial não efetua cálculos, apenas mostra a integral no formato usual o que, em determinadas situações, pode ser bastante útil.

Exemplos:

>int(2\*x,x);

$$x^2$$

>int(2\*x\*y,x);

$$x^2 y$$

>int(2\*x\*y,y);

$$x y^2$$

### 5.2. Integrais definidas e impróprias

Uma integral definida em um intervalo  $[a, b]$  é calculada com um comando do tipo  $\text{int}(f(x), x=a..b)$ . Integrais impróprias são fornecidas como integrais definidas. Nesses casos, podemos ter  $a$  ou  $b$  iguais a  $+\infty$  ou  $-\infty$ .

>Int(x/(x^2+16),x=a..0);

$$\int_a^0 \frac{x}{x^2 + 16} dx$$

>int(x/(x^2+16),x=a..0);

$$-\frac{1}{2} \ln(a^2 + 16) + 2 \ln(2)$$

>Int(exp(-2\*t)\*t^2\*ln(t),t=-infinity..0);

$$\int_{-\infty}^0 e^{(-2t)} t^2 \ln(t) dt$$

>int(exp(-2\*t)\*t^2\*ln(t),t=-infinity..0);

$$\infty$$

### 5.3. Integrais duplas e triplas

Podemos calcular uma integral dupla da seguinte forma:  $\text{Int}(\text{Int}(f(x,y), x=a..b), y=c..d)$ .

Exemplo:

```
>(Int(Int(2*x,x=-2..5),x=0..6));
```

$$2 \int_0^6 \int_{-2}^5 x \, dx \, dx$$

```
>int(int(2*x,x=-2..5),x=0..6);
```

126

A forma inercial da integral dupla de uma função de duas variáveis, definida por uma expressão algébrica  $f(x,y)$  nas variáveis  $x, y, z$ , no Maple, corresponde ao comando:  $\text{Doubleint}(f(x,y), x=a..b, y=c..d)$ , onde  $a..b$  e  $c..d$  denotam a variação do  $x$  e do  $y$ , respectivamente.

```
>with(student):
```

```
>Doubleint(x^2+2*y,x=y..3*y,y=1..2):%=value(%);
```

$$\int_1^2 \int_y^{3y} x^2 + 2y \, dx \, dy = \frac{251}{6}$$

Analogamente, a forma inercial da integral tripla de uma função de três variáveis, definida por uma expressão algébrica  $f(x,y,z)$  nas variáveis  $x, y, z$ , é dada por:  $\text{Tripleint}(f(x,y,z), x=a..b, y=c..d, z=e..f)$ , onde  $a..b$ ,  $c..d$  e  $e..f$  denotam a variação do  $x$ ,  $y$  e do  $z$ , respectivamente.  $\text{Tripleint}$  é equivalente a três comandos  $\text{Int}$  encaixados:  $\text{Int}(\text{Int}(\text{Int}(f(x,y,z), x=a..b), y=c..d), z=e..f)$ .

```
>Tripleint(x*y*z,x,y,z):%=value(%);
```

$$\iiint x y z \, dx \, dy \, dz = \frac{x^2 y^2 z^2}{8}$$

```
>Tripleint(x*y*z, z=0..sqrt(4-y^2), y=0..2*x, x=0..3):%=value(%);
```

$$\int_0^3 \int_0^{2x} \int_0^{\sqrt{4-y^2}} x y z \, dz \, dy \, dx = -162$$

## BIBLIOGRAFIA

ANDRADE, L. N.; Introdução à computação algébrica com o Maple. Rio de Janeiro: Sociedade Brasileira de Matemática, 2004.

PORTUGAL, R.; Introdução ao Maple. Petrópolis - RJ, 2002.